NAG Fortran Library Routine Document

F07MEF (SSYTRS/DSYTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07MEF (SSYTRS/DSYTRS) solves a real symmetric indefinite system of linear equations with multiple right-hand sides, $AX = B$, where A has been factorized by F07MDF (SSYTRF/DSYTRF).

2 Specification

```
SUBROUTINE F07MEF(UPLO, N, NRHS, A, LDA, IPIV, B, LDB, INFO)
ENTRY      ssytrs  (UPLO, N, NRHS, A, LDA, IPIV, B, LDB, INFO)
INTEGER            N, NRHS, LDA, IPIV(*), LDB, INFO
real               A(LDA,*), B(LDB,*)
CHARACTER*1        UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

To solve a real symmetric indefinite system of linear equations $AX = B$, this routine must be preceded by a call to F07MDF (SSYTRF/DSYTRF) which computes the Bunch–Kaufman factorization of A.

If UPLO = 'U', $A = PUDU^TP^T$, where P is a permutation matrix, U is an upper triangular matrix and D is a symmetric block diagonal matrix with 1 by 1 and 2 by 2 blocks; the solution X is computed by solving $PUDY = B$ and then $U^TP^TX = Y$.

If UPLO = 'L', $A = PLDL^TP^T$, where L is a lower triangular matrix; the solution X is computed by solving $PLDY = B$ and then $L^TP^TX = Y$.

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER*1 *Input*

On entry: indicates how A has been factorized as follows:

if UPLO = 'U', $A = PUDU^TP^T$, where U is upper triangular;

if UPLO = 'L', $A = PLDL^TP^T$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides.

Constraint: NRHS ≥ 0.

4: A(LDA,*) – ***real*** array *Input*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: details of the factorization of A, as returned by F07MDF (SSYTRF/DSYTRF).

5: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07MEF (SSYTRS/DSYTRS) is called.

Constraint: LDA $\geq \max(1, \mathrm{N})$.

6: IPIV($*$) – INTEGER array *Input*

Note: the dimension of the array IPIV must be at least $\max(1, \mathrm{N})$.

On entry: details of the interchanges and the block structure of D, as returned by F07MDF (SSYTRF/DSYTRF).

7: B(LDB,*) – ***real*** array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

On entry: the n by r right-hand side matrix B.

On exit: the n by r solution matrix X.

8: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07MEF (SSYTRS/DSYTRS) is called.

Constraint: LDB $\geq \max(1, \mathrm{N})$.

9: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b, the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

$|E| \leq c(n)\epsilon P|U|\,|D|\,|U^T|P^T$, if UPLO = 'U',

$|E| \leq c(n)\epsilon P|L|\,|D|\,|L^T|P^T$, if UPLO = 'L',

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

If $\hat{x}$ is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x-\hat{x}\|_\infty}{\|x\|_\infty} \leq c(n)\,\mathrm{cond}(A,x)\epsilon$$

where $\mathrm{cond}(A,x) = \||A^{-1}|\,|A|\,|x|\|_\infty / \|x\|_\infty \leq \mathrm{cond}(A) = \||A^{-1}|\,|A|\|_\infty \leq \kappa_\infty(A)$. Note that $\mathrm{cond}(A,x)$ can be much smaller than $\mathrm{cond}(A)$.

Forward and backward error bounds can be computed by calling F07MHF (SSYRFS/DSYRFS), and an estimate for $\kappa_\infty(A)$ $(= \kappa_1(A))$ can be obtained by calling F07MGF (SSYCON/DSYCON).

8 Further Comments

The total number of floating-point operations is approximately $2n^2r$.

This routine may be followed by a call to F07MHF (SSYRFS/DSYRFS) to refine the solution and return an error estimate.

The complex analogues of this routine are F07MSF (CHETRS/ZHETRS) for Hermitian matrices and F07NSF (CSYTRS/ZSYTRS) for symmetric matrices.

9 Example

To solve the system of equations $AX = B$, where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} -9.50 & 27.85 \\ -8.38 & 9.90 \\ -6.07 & 19.25 \\ -0.96 & 3.93 \end{pmatrix}.$$

Here A is symmetric indefinite and must first be factorized by F07MDF (SSYTRF/DSYTRF).

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07MEF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LDA, LWORK, NRHMAX, LDB
      PARAMETER        (NMAX=8,LDA=NMAX,LWORK=64*NMAX,NRHMAX=NMAX,
     +                 LDB=NMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N, NRHS
      CHARACTER        UPLO
*     .. Local Arrays ..
      real             A(LDA,NMAX), B(LDB,NRHMAX), WORK(LWORK)
      INTEGER          IPIV(NMAX)
*     .. External Subroutines ..
      EXTERNAL         ssytrf, ssytrs, X04CAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07MEF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, NRHS
      IF (N.LE.NMAX .AND. NRHS.LE.NRHMAX) THEN
*
*        Read A and B from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
```

```
            READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
         END IF
         READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*        Factorize A
*
         CALL ssytrf(UPLO,N,A,LDA,IPIV,WORK,LWORK,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.EQ.0) THEN
*
*           Compute solution
*
            CALL ssytrs(UPLO,N,NRHS,A,LDA,IPIV,B,LDB,INFO)
*
*           Print solution
*
            IFAIL = 0
            CALL X04CAF('General',' ',N,NRHS,B,LDB,'Solution(s)',IFAIL)
         ELSE
            WRITE (NOUT,*) 'The factor D is singular'
         END IF
      END IF
      STOP
*
      END
```

9.2 Program Data

```
F07MEF Example Program Data
  4  2                        :Values of N and NRHS
  'L'                         :Value of UPLO
  2.07
  3.87  -0.21
  4.20   1.87   1.15
 -1.15   0.63   2.06  -1.81   :End of matrix A
 -9.50  27.85
 -8.38   9.90
 -6.07  19.25
 -0.96   3.93                 :End of matrix B
```

9.3 Program Results

```
 F07MEF Example Program Results

 Solution(s)
             1          2
 1     -4.0000     1.0000
 2     -1.0000     4.0000
 3      2.0000     3.0000
 4      5.0000     2.0000
```